# $Id: PKGBUILD 84150 2013-02-12 01:00:22Z bgyorgy $
# Maintainer: Balló György <ballogyor+arch at gmail dot com>

pkgname=onboard
pkgver=0.98.2
pkgrel=1
pkgdesc="On-screen keyboard useful on tablet PCs or for mobility impaired users"
arch=('i686' 'x86_64')
url="https://launchpad.net/onboard"
license=('GPL')
depends=('python-dbus' 'python-gobject' 'gtk3' 'python-virtkey' 'librsvg' 'gsettings-desktop-schemas' 'desktop-file-utils' 'dconf' 'hicolor-icon-theme' 'xdg-utils')
makedepends=('python-distutils-extra')
optdepends=('at-spi2-atk: auto-show when editing text'
            'mousetweaks: hover click with mouse')
options=('!emptydirs')
install=$pkgname.install
source=(http://launchpad.net/$pkgname/${pkgver%.*}/$pkgver/+download/$pkgname-$pkgver.tar.gz)
md5sums=('7b44ab2873ca7b76ff1d9ab0b3398487')

build() {
  cd "$srcdir/$pkgname-$pkgver"

  python3 setup.py build
}

package() {
  cd "$srcdir/$pkgname-$pkgver"

  python3 setup.py install --root=$pkgdir/ --optimize=1
}
